Université PARIS-XII — Faculté de Sciences économiques et de gestion
Master *Sciences économiques*
Mention *Économie quantitative et expertise internationale*
Économétrie III — *F. LEGENDRE* — année universitaire 2005-06

Fiche de travail n°4
L'estimateur des MCQG

Je vous propose dans cette quatrième fiche de travail de programmer, à l'aide de SAS, l'estimateur des Moindres Carrés Quasi Généralisés et d'estimer en guise d'illustration une fonction de production «en taux de croissance» en utilisant les mêmes données que dans la fiche de travail n°2.

1 L'estimateur des MCQG

Rappelons, tout d'abord, que la transformation des MCG, si l'on a fait l'hypothèse d'erreurs composées, est la suivante :

$$\tilde{z}_{it} = z_{it} - (1-\sqrt{\theta})\overline{z}_i$$

où θ est égal à $\sigma_\nu^2/(T\sigma_\mu^2+\sigma_\nu^2)$. Il nous faut donc, pour mettre en œuvre les MCQG, détenir une estimation convergente, dans une première étape, de θ.

Nous allons faire simple en déduisant un estimateur à partir des estimations conduites dans les dimensions intra- et inter-individuelle. Notons, respectivement, $\underline{\widehat{u}}_W$ et $\underline{\widehat{u}}_B$ les vecteurs des résidus estimés des estimations intra- et inter-individuelle. Ces deux vecteurs sont de taille $NT \times 1$; dans la dimension inter, les moyennes individuelles sont donc répétées T fois. Notons, de plus, $SCR_W = \underline{\widehat{u}}_W'\underline{\widehat{u}}_W$ et $SCR_B = \underline{\widehat{u}}_B'\underline{\widehat{u}}_B$. On a

$$E(SCR_W) = \sigma_\nu^2[NT-N-k]$$

où k est le nombre de variables explicatives du modèle. De même, on a

$$E(SCR_B) = (T\sigma_\mu^2+\sigma_\nu^2)(N-k)$$

L'estimateur suivant de θ est alors habituellement proposé :

$$\hat{\theta} = \frac{SCR_W/[NT-N-k]}{SCR_B/(N-k)}$$

C'est donc cette grandeur qu'il faut dans un premier temps calculer.

Dans une seconde étape, il faut transformer les données en utilisant $\hat{\theta}$:

$$z_{it}^\star = z_{it} - (1-\sqrt{\hat{\theta}})\overline{z}_i$$

Le modèle devient ainsi

$$\underline{y}^\star = X^\star\underline{a} + \underline{u}^\star$$

L'estimateur des MCQG du vecteur $\underline{a}$ est l'estimateur des MCO de ce modèle :

$$\underline{\widehat{a}}_{MCQG} = (X^{\star\prime}X^\star)^{-1}X^{\star\prime}\underline{y}^\star$$

2 Le programme SAS

Ci-après figure le programme SAS qui met en œuvre l'estimateur de MCQG.

```
PROC reg DATA = final OUTEST = estimations EDF ;
   MODEL dlqi = dlki dlli / NOINT ;
   MODEL dlqb = dlkb dllb ;
RUN ;
DATA mcqg ;
  IF _N_ EQ 1 THEN DO ;
    SET moyennes (DROP = _TYPE_) NOBS = n ;
    SET estimations (DROP = _TYPE_) ;
    nt = _EDF_ + _P_ ;
    scr_w = _RMSE_*_RMSE_*_EDF_ ;
    SET estimations (DROP = _TYPE_ FIRSTOBS = 2) ;
    scr_b = _RMSE_*_RMSE_*_EDF_ ;
    theta_chapeau = (scr_w/(nt-n)) / (scr_b/n) ;
    RETAIN theta_chapeau ;
    PUT theta_chapeau = ;
  END ;
  SET final ;
  dlq_star = dlq - (1-SQRT(theta_chapeau))*dlqb ;
  dlk_star = dlk - (1-SQRT(theta_chapeau))*dlkb ;
  dll_star = dll - (1-SQRT(theta_chapeau))*dllb ;
RUN ;
PROC reg DATA = mcqg ;
  MODEL dlq_star = dlk_star dll_star ;
RUN ;
```

Vous pouvez noter les points suivants.

1. L'utilisation du mot-clé `EDF` dans la `PROC reg` pour disposer, dans la table repérée par le mot-clé `OUTEST`, des variables `_EDF_` (le nombre d'observations moins le nombre de variables explicatives) et `_P_` (le nombre de variables explicatives).
2. L'utilisation du mot-clé `NOBS` sur la table `moyennes` pour récupérer le nombre d'individus.
3. L'utilisation du mot-clé `FIRSTOBS` en option d'une table pour spécifier la première observation à lire.
4. La nécessité d'utiliser l'instruction `RETAIN` pour éviter que la variable `theta_chapeau` ne soit remise à valeur manquante au début de chaque itération de l'étape `DATA`.
5. Par paresse, l'omission de la correction, dans le calcul de $\hat{\theta}$, par le nombre de variables explicatives du modèle (la grandeur notée k).